Komparative Statik

1. Definition

Aufgabe: Definieren Sie den Begriff der komparativen Statik.

Definition: Komparative Statik untersucht die Veränderung ökonomischer Variablen in Abhängigkeit von Änderungen ökonomischer Parameter.
=> Differenzieren als Hauptwerkzeug

Zwei Arten:
Explizites Differenzieren: abzuleitende Funktion ist explizit gegeben
=> kann direkt abgeleitet werden
Implizites Differenzieren: abzuleitende Funktion ist implizit durch eine Gleichung gegeben
=> muss implizit abgeleitet werden

2. Komparative Statik Explizit:

Aufgabe: Führen Sie die komparative Statik für die Ergebnisse der Aufgaben 1.2 a) und b) durch.

Rekapitulation für Aufgabe 1.2 a) + Komp. Statik.

Rekapitulation:

Gegeben:

In[∘]:=
```
xd = b - a * p;
xs = c * p;
```

Gesucht: Gleichgewichtspreis u. Menge

Lösung:
1. Angebot u. Nachfrage gleichsetzen

In[∘]:=
```
gg = (xd == xs);
```

2. Lösen die Gleichgewichtsbedingung nach dem Preis auf

In[∘]:=
```
sol = Solve[gg, p][[1, 1]];
pgg = p /. sol
```

Out[∘]= $\frac{b}{a + c}$

3. Setzen Preis in Angebot oder Nachfrage ein (im GG sowieso gleich)

In[∘]:=
```
xgg = ((xd /. sol) // Simplify)
```
Out[∘]= $\frac{b\,c}{a + c}$

Komparative Statik

Welche Parameter gibt es?

In[∘]:=
```
parameter = {a, b, c};
```

Leiten nun Preis u. Menge nach den Parametern ab
(Ableitungen sind in Reihenfolge der Parameter; Zuerst Preis, dann Menge;)

In[∘]:=
```
pggd = D[pgg, {parameter}]
xggd = D[xgg, {parameter}] // Simplify
```
Out[∘]= $\left\{-\frac{b}{(a + c)^2}, \frac{1}{a + c}, -\frac{b}{(a + c)^2}\right\}$

Out[∘]= $\left\{-\frac{b\,c}{(a + c)^2}, \frac{c}{a + c}, \frac{a\,b}{(a + c)^2}\right\}$

Interpretation

Lokale Änderungen sind annäherend Linear, d.h. die Änderung der Abhängigen ist annähernd proportional zur Änderung des Parameters.

Beispiel: Graph des Gleichgewichtspreises in Abhängigkeit von Parameter a und dessen Tangente im Punkt a=1

In[∘]:=
```
exP = {b → 1, c → 1};
ex2P = {a → 1, b → 1, c → 1};
tangente = (pggd[[1]] /. ex2P) * (a - 1) + (pgg /. ex2P)
Plot[{pgg /. ex, tangente}, {a, 0, 2}]
```
Out[∘]= $\frac{1}{2} + \frac{1 - a}{4}$

Out[∘]=


Anhand der Vorzeichen der Ableitungen kann man sagen, ob die Abhängige Variable lokal größer oder kleiner wird, wenn die Änderung des Parameters positiv oder negativ ist.
Allgemein: dy = f’(x_0) dx

Bsp.:

$\left(-\frac{b}{(a+c)^2}\right) > 0$, da a,b,c >0

Also, wenn da > 0 , dann ist

$$\boxed{dp == \left(-\frac{b}{(a+c)^2}\right) da < 0}$$

Analog für die anderen Parameter.

Hausaufgabe: Untersuchen Sie die Vorzeichen einiger selbstausgewählter Änderungen von Preis- oder Menge als Reaktion auf Parameteränderungen.

Rekapitulation für Aufgabe 1.2 b) + Komp. Statik.

Rekapitulation

Gegeben:

```
In[•]:= xd = b - a * p;
        cost = c * x ^ 2;
```

Gesucht: Unternehmensangebot und Marktgleichgewicht

Lösung:

1. Unternehmen maximiert Profit = Einnahmen - Ausgaben

```
In[•]:= profit = x * p - cost
```

Out[•]= $p\,x - c\,x^2$

2. Ableiten nach Angebotsmenge (Preisnehmer!) u. gleich 0 setzen

```
In[•]:= foc = (D[profit, x] == 0)
```

Out[•]= $p - 2\,c\,x == 0$

3. Lösen nach Menge auf

```
In[•]:= solS = Solve[foc, x][[1, 1]]
        xs = x /. solS;
```

Out[•]= $x \to \frac{p}{2\,c}$

4. Setzen Angebot gleich Nachfrage, lösen nach Preis auf u. setzen in Angebots- oder Nachfragefunktion

```
In[•]:= sol = Solve[(xd == xs), p][[1, 1]]
        pgg = p /. sol;
        xgg = (xd /. sol) // Simplify
```

Out[•]= $p \to \frac{2\,b\,c}{1 + 2\,a\,c}$

Out[•]= $\frac{b}{1 + 2\,a\,c}$

5. Ableitungen nach den Parametern

In[•]:=
```
parameter = {a, b, c};
pggd = D[pgg, {parameter}] // Simplify
xggd = D[xgg, {parameter}]
```

Out[•]= $\left\{-\frac{4\,b\,c^2}{(1+2\,a\,c)^2},\ \frac{2\,c}{1+2\,a\,c},\ \frac{2\,b}{(1+2\,a\,c)^2}\right\}$

Out[•]= $\left\{-\frac{2\,b\,c}{(1+2\,a\,c)^2},\ \frac{1}{1+2\,a\,c},\ -\frac{2\,a\,b}{(1+2\,a\,c)^2}\right\}$

Es gilt bspw. für die Preisänderung in Abhängigkeit von einer Änderung von Parameter a

$$dp == \left(-\frac{4\,b\,c^2}{(1+2\,a\,c)^2}\right) * da$$

In[•]:= $\text{Assuming}\left[\{a>0,\ b>0,\ c>0,\ da>0\},\ \text{Simplify}\left[\left(-\frac{4\,b\,c^2}{(1+2\,a\,c)^2}\right) * da < 0\right]\right]$

Out[•]= True

3. Komparative Statik Implizit:

Führen Sie die komparative Statik implizit an- hand der Optimalitätsbedingungen folgender Optimierungsaufgaben durch

Aufgabe a)

Ein Unternehmen produziert mit zwei Produktionsfaktoren f1, f2 zu Einkaufspreisen w1,w2, zwei Güter x1(f1) = ln(f1), x2(f2) = ln(f2) zu Absatzpreisen p1, p2.

Gegeben:

Produktionsfunktionen:

In[•]:=
```
x1 = Log[f1]
x2 = Log[f2]
```

Out[•]= Log[f1]

Out[•]= Log[f2]

Preise (Parameter):

Absatz: p1, p2

Einkauf: w1, w2

Gesucht: Angebotsänderungen in Abhängigkeit der Preisänderungen

Lösung:

1. Stellen Gewinnfunktion des Unternehmens auf

In[•]:=
```
profit  = x1 * p1 - w1 * f1 + x2 * p2 - w2 * f2
```

Out[•]= - f1 w1 - f2 w2 + p1 Log[f1] + p2 Log[f2]

2. Stellen Optimalitätsbedingungen auf (nach Angebotsmengen ableiten u. gleich 0 setzen)

In[•]:=
```
(# == 0) & /@ (D[profit, {{f1, f2}}])
```

Out[•]=
$$\left\{\frac{p1}{f1} - w1 == 0,\ \frac{p2}{f2} - w2 == 0\right\}$$

3. Um die implizite Ableitungen zu bestimmen bspw. nach w1, behandeln wir nun f1, f2 als Abhängige von w1. FOC sehen dann folgendermaßen aus:

In[•]:=
```
(# == 0) & /@ ((D[profit, {{f1, f2}}]) /. {f1 → f1[w1], f2 → f2[w1]})
```

Out[•]=
$$\left\{-w1 + \frac{p1}{f1[w1]} == 0,\ -w2 + \frac{p2}{f2[w1]} == 0\right\}$$

4. Leiten die linke Seiten nun nach w1 ab

In[•]:=
```
eq = ((# == 0) & /@ D[((D[profit, {{f1, f2}}]) /. {f1 → f1[w1], f2 → f2[w1]}), w1])
```

Out[•]=
$$\left\{-1 - \frac{p1\ f1'[w1]}{f1[w1]^2} == 0,\ -\frac{p2\ f2'[w1]}{f2[w1]^2} == 0\right\}$$

5. Lösen das Gleichungssystem schließlich nach den Ableitungen von f1, f2 auf (f1′[w1] bzw. f2′[w1])

In[•]:=
```
Solve[eq, { f1′[w1], f2′[w1]}][[1]]
```

Out[•]=
$$\left\{f1'[w1] \to -\frac{f1[w1]^2}{p1},\ f2'[w1] \to 0\right\}$$

6. D.h., wenn dw1 >0 dann ist df1 < 0 und df2 = 0

Insgesamt für alle Parameter (In Reihenfolge der Parameter)

In[•]:=
```
parameter = {w1, w2, p1, p2};
implDiff[para_] :=
 Solve[((# == 0) & /@ D[((D[profit, {{f1, f2}}]) /. {f1 → f1[para], f2 → f2[para]}),
      para]), { f1′[para], f2′[para]}][[1]]
(implDiff[#]) & /@ parameter
```

Out[•]=
$$\left\{\left\{f1'[w1] \to -\frac{f1[w1]^2}{p1},\ f2'[w1] \to 0\right\},\ \left\{f1'[w2] \to 0,\ f2'[w2] \to -\frac{f2[w2]^2}{p2}\right\},\ \left\{f1'[p1] \to \frac{f1[p1]}{p1},\ f2'[p1] \to 0\right\},\ \left\{f1'[p2] \to 0,\ f2'[p2] \to \frac{f2[p2]}{p2}\right\}\right\}$$

Rekapitulation Aufgabe 2 u. komparative Statik implizit

Rekapitulation

Gegeben: Monopolistischer Markt mit

```
xd = b - a * p
cost = c * x ^ 2
```

Gesucht: Bedingungen erster Ordnung für das Unternehmensangebot (mehr wird für die implizite komparative Statik nicht benötigt)

Lösung:

1. Berechnen inverse Nachfrage (lösen Nachfrage nach p auf)

In[∘]:=
```
pd = p /. Solve[xd == x, p][[1]]
```

Out[∘]= $\frac{b - x}{a}$

2. Stellen Profitfunktion des Unternehmens auf

In[∘]:=
```
profit = pd * x - cost
```

Out[∘]= $\frac{(b - x)\, x}{a} - c\, x^2$

3. Leiten nach Angebotspreis ab und setzen gleich 0

In[∘]:=
```
D[profit, x] == 0
```

Out[∘]= $\frac{b - x}{a} - \frac{x}{a} - 2\, c\, x == 0$

Komparative Statik

1. Wir behandeln nun wieder x als Abhängig von einem Parameter bspw. Parameter a

In[∘]:=
```
D[profit, x] /. {x → x[a]}
```

Out[∘]= $\frac{b - x[a]}{a} - \frac{x[a]}{a} - 2\, c\, x[a]$

2. Leiten den Gewinn nun nach a ab und setzen gleich 0

In[∘]:=
```
eq = (D[D[profit, x] /. {x → x[a]}, a] == 0)
```

Out[∘]= $-\frac{b - x[a]}{a^2} + \frac{x[a]}{a^2} - \frac{2\, x'[a]}{a} - 2\, c\, x'[a] == 0$

3. Lösen die Gleichung nun nach der Ableitung von x nach a $x'[a]$ ab

In[∘]:=
```
Solve[eq, x'[a]][[1]]
```

Out[∘]= $\left\{x'[a] \to \frac{-b + 2\, x[a]}{2\, a\, (1 + a\, c)}\right\}$

4. Können nun versuchen Vorzeichen der Angebotsänderung bestimmen. Ansatz:

```
dx = x'[a] * da
```

Die impliziten Ableitungen für alle Parameter

In[∘]:=
```
parameter = {a, b, c}
implDiff[para_] :=
 Solve[(D[D[profit, x] /. {x → x[para]}, para] == 0), x'[para]][[1]]
(implDiff[#]) & /@ parameter
```

Out[∘]= {a, b, c}

Out[∘]= $\left\{\left\{x'[a] \to \frac{-b + 2\, x[a]}{2\, a\, (1 + a\, c)}\right\}, \left\{x'[b] \to \frac{1}{2\, (1 + a\, c)}\right\}, \left\{x'[c] \to -\frac{a\, x[c]}{1 + a\, c}\right\}\right\}$